```
Manipulate[
 Grid[{{"Old", opr, "New"},
   If[style == "Hollow", {Rep1[OldColor], Magnify["⟹", 3], Rep1[Color]},
    {Rep2[OldColor], Magnify["⟹", 3], Rep2[Color]}]}],
 Grid[{
   {Style["Pocket Cube with Hint", Bold, Large, Blue],
    SpanFromLeft, SpanFromLeft, SpanFromLeft},
   {"-----------", "-----------", "-----------", "-----------"},
   {"", Style["UP", Medium], Style["RIGHT", Medium], Style["FRONT", Medium]},
   {Style[" ↻", Bold, Medium],
    Button[Mouseover["U", Rotate["U", -π/6]], {OldColor = Color;
      Operations["U"];
      opr = "U";
      seq = "u" <> seq}],
    Button[Mouseover["R", Rotate["R", -π/6]], {OldColor = Color;
      Operations["R"];
      opr = "R";
      seq = "r" <> seq}],
    Button[Mouseover["F", Rotate["F", -π/6]], {OldColor = Color;
      Operations["F"];
      opr = "F";
      seq = "f" <> seq}]},
   {Style["↺", Bold, Medium],
    Button[Mouseover["u", Rotate["u", π/6]], {OldColor = Color;
      Operations["u"];
      opr = "u";
      seq = "U" <> seq}],
    Button[Mouseover["r", Rotate["r", π/6]], {OldColor = Color;
      Operations["r"];
      opr = "r";
      seq = "R" <> seq}],
    Button[Mouseover["f", Rotate["f", π/6]], {OldColor = Color;
      Operations["f"];
      opr = "f";
      seq = "F" <> seq}]},
   {PopupMenu[Dynamic[style], {"Hollow", "Solid"}, "Style"], Button["Reset", Reset],
    Button["Shuffle", Shuffle], Button["Recover", If[seq != "",
      {OldColor = Color;
       Operations[StringTake[seq, 1]];
       opr = StringTake[seq, 1];
       seq = StringDrop[seq, 1];}, Reset]]},
   {Style["Hint", Bold, FontSize → 18, Red],
    Dynamic[If[seq != "", Style[StringReplace[seq = Reduction[seq],
        {"UU" → "U²", "RR" → "R²", "FF" → "F²"}], Bold, FontSize -> 18],
      Style["Recovered!", Darker[Green], Bold, FontSize → 18]]], SpanFromLeft}
```

```
  }],

 Initialization :> {
   Colors = {Glow[Red], Glow[Yellow], Glow[Lighter[Blue]],
     Glow[Lighter[Purple, 0.7]], Glow[Orange], Glow[Green]};
   (*customize colors*)
   FaceIndices = PolyhedronData["Cube", "FaceIndices"];
VertexCoords =
    Flatten[Table[{x, y, z}, {x, -1, 1, 2}, {y, -1, 1, 2}, {z, -1, 1, 2}], 2];
   (*the coordinates of the 8 vertexes*)
SubFaceCoords = 0.9 Table[VertexCoords[[ FaceIndices[[f, v]] ]] , {f, 6}, {v, 4}];
   (*the 4 coodinates of sub squares of each face*)
   Face0 =
    Table[
     Polygon[Table[
       VertexCoords[[ FaceIndices[[f, s]] ]] + SubFaceCoords[[f, v]],
       {v, 4}]],
     {f, 6}, {s, 4}]; (*coordinates of the cube components*)
   Core = Table[{Glow[Black],
      Polygon[1.8 Table[VertexCoords[[ FaceIndices[[f, v]] ]] , {v, 4}]]}, {f, 6}];
   (*solid core of the cube*)
   Color0 = Table[ f, {f, 6}, {s, 4}]; (*initial color state*)
   Color = Color0; OldColor = Color;
   shuffle = {}; seq = ""; opr = "";

   Rep1[c_] := Graphics3D[(*representation type 1, hollow*)
     Transpose[{Colors[[Flatten@c]], Table[Opacity[0.9], {24}], Flatten@Face0}],
     Boxed → False, Lighting -> None, ViewPoint → {Pi, Pi/2, 2}];
   Rep2[c_] := Graphics3D[(*representation type 2, solid*)
     Append[Transpose[{Colors[[Flatten@c]], Flatten@Face0}], Core],
     Boxed → False, Lighting -> None, ViewPoint → {Pi, Pi/2, 2}];
   Operations[opr_] := (*Define the color transfomation of each operation.*)
    Which[
     opr == "U", (*up clockwise*)
     {Color[[1]] = RotateLeft[Color[[1]]];
      {Color[[4, 1]], Color[[4, 4]], Color[[3, 1]], Color[[3, 4]],
        Color[[2, 2]], Color[[2, 1]], Color[[6, 1]], Color[[6, 4]]} =
       RotateRight[{Color[[4, 1]], Color[[4, 4]], Color[[3, 1]], Color[[3, 4]],
         Color[[2, 2]], Color[[2, 1]], Color[[6, 1]], Color[[6, 4]]}, 2]},
     opr == "F", (*front clockwise*)
     {Color[[2]] = RotateLeft[Color[[2]]];
      {Color[[1, 4]], Color[[1, 1]], Color[[3, 1]], Color[[3, 2]],
        Color[[5, 3]], Color[[5, 4]], Color[[6, 3]], Color[[6, 4]]} =
       RotateRight[{Color[[1, 4]], Color[[1, 1]], Color[[3, 1]], Color[[3, 2]],
         Color[[5, 3]], Color[[5, 4]], Color[[6, 3]], Color[[6, 4]]}, 2]},
     opr == "R", (*right clockwise*)
     {Color[[3]] = RotateLeft[Color[[3]]];
      {Color[[1, 1]], Color[[1, 2]], Color[[4, 1]], Color[[4, 2]],
        Color[[5, 2]], Color[[5, 3]], Color[[2, 4]], Color[[2, 1]]} =
       RotateRight[{Color[[1, 1]], Color[[1, 2]], Color[[4, 1]], Color[[4, 2]],
         Color[[5, 2]], Color[[5, 3]], Color[[2, 4]], Color[[2, 1]]}, 2]},
     opr == "u" || opr == "D'", (*up anticlockwise*)
     Operations["U"]; Operations["U"]; Operations["U"],
     opr == "f" || opr == "B'", (*front anticlockwise*)
```

```
    Operations["F"]; Operations["F"]; Operations["F"],
    opr == "r" || opr == "L'", (*right anticlockwise*)
    Operations["R"]; Operations["R"]; Operations["R"]
   ];
  Oppo[opr_] := Which[opr == "U", "u", opr == "u", "U",
    opr == "F", "f", opr == "f", "F", opr == "R", "r", opr == "r", "R"];
  (*define the opposite operations*)
  Recover[shuffle_] := Table[Oppo[shuffle[[s]]], {s, Length[shuffle], 1, -1}];
  (*find the recovery operations *)
  Reduction[seq_] := StringReplace[seq,
    {"UUUU" → "", "UUU" → "u"(*,"UU"→"UU"*),
     "uuuu" → "", "uuu" → "U", "uu" → "UU", "Uu" -> "", "uU" → "",
     "RRRR" → "", "RRR" → "r"(*,"RR"→"RR"*), "rrrr" → "",
     "rrr" → "R", "rr" → "RR", "Rr" → "", "rR" → "",
     "FFFF" → "", "FFF" → "f"(*,"FF"→"FF"*), "ffff" → "",
     "fff" → "F", "ff" → "FF", "Ff" → "", "fF" → ""}];
  (*simplify the operations*)
  Shuffle := (Reset;
    shuffle = RandomChoice[{"U", "R", "F"}, RandomInteger[{10, 20}]];
    seq = StringJoin[Recover[shuffle]];
    Table[Operations[s], {s, shuffle}];
    OldColor = Color0;
    opr = "shuffle");
  Reset := (Color = Color0;
    OldColor = Color;
    opr = "";
    shuffle = {};
    seq = "");
 }
]
```

Pocket Cube with Hint

	UP	RIGHT	FRONT
↻	U	R	F
↺	u	r	f

Hollow | Reset | Shuffle | Recover

Hint

Recovered!

Old



New